P 10

PRÁCTICA-CAMPOS VECTORIALES

Ejercicio Propuesto P-10.1

Dado el campo vectorial

$$\vec{F}(x,y)=x*y\,\vec{i}+\left(x^2-y^2\right)\vec{j}$$

Obtener las líneas de corriente y dibujarlas conjuntamente con el campo vectorial

Solución P-10.1

★ Determinamos la E.D. de las líneas de corriente

En cada punto, el vector tangente a la curva que pasa por él será (m(x,y),n(x,y)) que será el campo de vectores y tiene pendiente n(x,y)/m(x,y)

```
m[x_, y_] = x * y;
n[x_, y_] = x^2 - y^2;
```

Definimos la E.D. de las líneas de corriente y la resolvemos

```
edlc = y'[x] == n[x, y[x]] / m[x, y[x]]
```

$$y'[x] == \frac{x^2 - y[x]^2}{x\, y[x]}$$

```
S = DSolve[edlc, y[x], x]
```

$$\left\{\left\{y[x] \to -\frac{\sqrt{x^4 + 2\,C[1]}}{\sqrt{2}\,x}\right\}, \left\{y[x] \to \frac{\sqrt{x^4 + 2\,C[1]}}{\sqrt{2}\,x}\right\}\right\}$$

Definamos dos funciones que representen la solución general de la E.D.O. dada a partir de la solución obtenida

```
s1[x_, c_] = S[[1, 1, 2]] /. C[1] → c
s2[x_, c_] = S[[2, 1, 2]] /. C[1] → c
```

$$-\frac{\sqrt{2\,c + x^4}}{\sqrt{2}\,x}$$

$$\frac{\sqrt{2\,c + x^4}}{\sqrt{2}\,x}$$

Obtengamos y dibujemos una familia de soluciones a partir de la siguiente lista

```
sol = Flatten[Table[{s1[x, c], s2[x, c]}, {c, 0.1, 10, 1}], 2]
```

$$\left\{-\frac{\sqrt{0.2 + x^4}}{\sqrt{2}\,x}, \frac{\sqrt{0.2 + x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{2.2 + x^4}}{\sqrt{2}\,x}, \frac{\sqrt{2.2 + x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{4.2 + x^4}}{\sqrt{2}\,x}, \frac{\sqrt{4.2 + x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{6.2 + x^4}}{\sqrt{2}\,x}, \frac{\sqrt{6.2 + x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{8.2 + x^4}}{\sqrt{2}\,x}, \frac{\sqrt{8.2 + x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{10.2 + x^4}}{\sqrt{2}\,x}, \frac{\sqrt{10.2 + x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{12.2 + x^4}}{\sqrt{2}\,x}, \frac{\sqrt{12.2 + x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{14.2 + x^4}}{\sqrt{2}\,x}, \frac{\sqrt{14.2 + x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{16.2 + x^4}}{\sqrt{2}\,x}, \frac{\sqrt{16.2 + x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{18.2 + x^4}}{\sqrt{2}\,x}, \frac{\sqrt{18.2 + x^4}}{\sqrt{2}\,x}\right\}$$

```
lineascorriente = Plot[Evaluate[sol], {x, -3, 3}, PlotRange → {-4, 4}]
```



En cada punto, el vector tangente a la curva que pasa por él será (m(x,y),n(x,y)) que será el campo de vectores.

```
campovectorial = VectorPlot[{m[x, y], n[x, y]}, {x, -3, 3}, {y, -4, 4}, Axes → True];
```

★ Dibujamos conjuntamente el campo vectorial y las líneas de corriente

```
Show[{campovectorial, lineascorriente}, PlotRange → {-4, 4}]
```



▼ Ejercicio Propuesto P-10.2

Dado el campo vectorial

$$\vec{F}(x,y)=y\,\vec{i}-x\,\vec{j}$$

a) Obtener el campo de vectores asociado y dibujarlo

b) Obtener la ecuación diferencial de las lineas de corriente y hallar la solución general

c) Obtener una familia de soluciones y dibujarla

d) Dibujar conjuntamente la familia de curvas y el campo vectorial

---

▼ Solución P-10.2

★ Apartado a

```
m[x_, y_] = y;
n[x_, y_] = -x;
```

```
campovectorial = VectorPlot[{m[x, y], n[x, y]}, {x, -1, 1}, {y, -1, 1}, Axes → True]
```



★ Apartado b

```
edlc = n[x, y[x]] / m[x, y[x]] == y'[x]
```

$$-\frac{x}{y[x]} == y'[x]$$

```
S = DSolve[edlc, y[x], x]
```

$$\left\{\left\{y[x] \to -\sqrt{-x^2 + 2\,C[1]}\right\}, \left\{y[x] \to \sqrt{-x^2 + 2\,C[1]}\right\}\right\}$$

```
s1[x_, c_] = S[[1, 1, 2]] /. C[1] → c / 2
s2[x_, c_] = S[[2, 1, 2]] /. C[1] → c / 2
```

$$-\sqrt{c - x^2}$$

$$\sqrt{c - x^2}$$

★ Apartado c

```
sol = Flatten[Table[{s1[x, c], s2[x, c]}, {c, 0.01, 1, .1}], 2]
```

$$\left\{-\sqrt{0.01 - x^2}, \sqrt{0.01 - x^2}, -\sqrt{0.11 - x^2}, \sqrt{0.11 - x^2}, -\sqrt{0.21 - x^2}, \sqrt{0.21 - x^2}, -\sqrt{0.31 - x^2}, \sqrt{0.31 - x^2}, -\sqrt{0.41 - x^2}, \sqrt{0.41 - x^2}, -\sqrt{0.51 - x^2}, \sqrt{0.51 - x^2}, -\sqrt{0.61 - x^2}, \sqrt{0.61 - x^2}, -\sqrt{0.71 - x^2}, \sqrt{0.71 - x^2}, -\sqrt{0.81 - x^2}, \sqrt{0.81 - x^2}, -\sqrt{0.91 - x^2}, \sqrt{0.91 - x^2}\right\}$$

```
familiasol = Plot[Evaluate[sol], {x, -1, 1}, AspectRatio → Automatic]
```



★ Apartado d

```
Show[{campovectorial, familiasol}]
```



Ejercicio Propuesto P-10.3

Dado el campo Vectorial $\vec{F}(x,y)=x\,\vec{i}+2\,y\,\vec{j}$

a) Obtener el campo de vectores asoiciado y dibujarlo

b) Obtener la ecuación diferencial de las lineas de corriente y hallar la solución general

c) Dibujar conjuntamente las las lineas de corriente y el campo vectorial

d) Obtener la E.D. de las trayectorias ortogonales

e) Dibujar conjuntamente la familia de curvas y el campo de vectores asociado a las trayectorias ortogonales

f) Dibujar conjuntamente las dos familias de curvas y los dos campos vectoriales asociados

Solución P-10.3

★ Apartado a

```
m[x_, y_] = x;
n[x_, y_] = 2 y;

campvecti = VectorPlot[{m[x, y], n[x, y]}, {x, -1.5, 1.5}, {y, -1, 1}, Axes → True]
```



★ Apartado b

```
edlc = n[x, y[x]] / m[x, y[x]] == y'[x]
```

$$\frac{2\,y[x]}{x} == y'[x]$$

```
Si = DSolve[edlc, y[x], x]
```

$\{\{y[x] \to x^2\,C[1]\}\}$

```
si[x_, c_] = Si[[1, 1, 2]] /. C[1] → c
```

$c\,x^2$

★ Apartado c

```
solti = Flatten[{Table[si[x, c], {c, -0.5, 0.5, .1}], Table[si[x, c], {c, -10, 10, 2}]}, 2]
```

$\{-0.5\,x^2, -0.4\,x^2, -0.3\,x^2, -0.2\,x^2, -0.1\,x^2, 2.77556\times10^{-17}\,x^2, 0.1\,x^2, 0.2\,x^2, 0.3\,x^2, 0.4\,x^2, 0.5\,x^2, -10\,x^2, -8\,x^2, -6\,x^2, -4\,x^2, -2\,x^2, 0, 2\,x^2, 4\,x^2, 6\,x^2, 8\,x^2, 10\,x^2\}$

```
famsolti = Plot[Evaluate[solti], {x, -1.5, 1.5},
  PlotRange → {-1, 1}, AspectRatio → Automatic, Ticks → None]
```



```
Show[{campvecti, famsolti}]
```



★ Apartado d

```
edto = y'[x] == -m[x, y[x]] / n[x, y[x]]
```

$$y'[x] == -\frac{x}{2\,y[x]}$$

```
Sto = DSolve[edto, y[x], x]
```

$$\left\{\left\{y[x] \to -\frac{\sqrt{-x^2 + 4\,C[1]}}{\sqrt{2}}\right\}, \left\{y[x] \to \frac{\sqrt{-x^2 + 4\,C[1]}}{\sqrt{2}}\right\}\right\}$$

```
sto1[x_, c_] = Sto[[1, 1, 2]] /. C[1] → c / 4
sto2[x_, c_] = Sto[[2, 1, 2]] /. C[1] → c / 4
```

$$-\frac{\sqrt{c - x^2}}{\sqrt{2}}$$

$$\frac{\sqrt{c - x^2}}{\sqrt{2}}$$

★ Apartado e

```
solto = Flatten[Table[{sto1[x, c], sto2[x, c]}, {c, 0.2, 2, .2}], 2]
```

$$\left\{-\frac{\sqrt{0.2 - x^2}}{\sqrt{2}}, \frac{\sqrt{0.2 - x^2}}{\sqrt{2}}, -\frac{\sqrt{0.4 - x^2}}{\sqrt{2}}, \frac{\sqrt{0.4 - x^2}}{\sqrt{2}}, -\frac{\sqrt{0.6 - x^2}}{\sqrt{2}}, \frac{\sqrt{0.6 - x^2}}{\sqrt{2}}, -\frac{\sqrt{0.8 - x^2}}{\sqrt{2}}, \frac{\sqrt{0.8 - x^2}}{\sqrt{2}}, -\frac{\sqrt{1. - x^2}}{\sqrt{2}}, \frac{\sqrt{1. - x^2}}{\sqrt{2}}, -\frac{\sqrt{1.2 - x^2}}{\sqrt{2}}, \frac{\sqrt{1.2 - x^2}}{\sqrt{2}}, -\frac{\sqrt{1.4 - x^2}}{\sqrt{2}}, \frac{\sqrt{1.4 - x^2}}{\sqrt{2}}, -\frac{\sqrt{1.6 - x^2}}{\sqrt{2}}, \frac{\sqrt{1.6 - x^2}}{\sqrt{2}}, -\frac{\sqrt{1.8 - x^2}}{\sqrt{2}}, \frac{\sqrt{1.8 - x^2}}{\sqrt{2}}, -\frac{\sqrt{2. - x^2}}{\sqrt{2}}, \frac{\sqrt{2. - x^2}}{\sqrt{2}}\right\}$$

```
famsolto = Plot[Evaluate[solto], {x, -1.5, 1.5}, PlotRange → {-1, 1}]
```



```
campvecto = VectorPlot[{2 y, -x}, {x, -1.5, 1.5}, {y, -1, 1}, AspectRatio → Automatic]
```




```
Show[{campvecto, famsolto}, PlotRange → {-1, 1}, Ticks → None, AspectRatio → Automatic]
```



★ Apartado f

```
Show[{famsolto, famsolti}, PlotRange → {-1, 1}, Ticks → None]
```



```
Show[{campvecto, campvecti}, PlotRange → {-1, 1}, AspectRatio → Automatic]
```



```
Show[{campvecto, campvecti, famsolto, famsolti},
 PlotRange → {-1, 1}, AspectRatio → Automatic]
```